

Mapping Roanoke Island: From 1585 to Present

Barbara B. Okerson, Anthem, Inc.

ABSTRACT

One of the first maps of the present United States was John White's 1585 map of the Albemarle Sound and Roanoke Island, the site of the Lost Colony and the site of my present home. This presentation looks at advances in mapping through the ages, from the early surveys and hand-painted maps, through lithographic and photochemical processes, to digitization and computerization. Inherent difficulties in including small pieces of coastal land—often removed from map boundary files and data sets to smooth a boundary—are also discussed. The paper concludes with several current maps of Roanoke Island created with SAS®.

INTRODUCTION

Geospatial maps are the dominant means of recording and communicating information about spatial characteristics of humanity and civilization. Maps are produced and used throughout the world by scientists, scholars, governments, businesses, and ordinary people to explore their environmental, economic, political, social and spatial needs. For data analysts maps serve four primary purposes:

- Set boundaries;
- Recording and storing spatial information at a set point in time;
- Visually analyzing spatial distributions and patterns, and
- Communicating spatial information.

Most data have spatial aspects. A map represents an area of land or sea showing displaying physical and/or demographic features within spatial or geographic boundaries. Spatial or geospatial data represents location, size and shape where geographic data can refer only to the location. Because most areas of the world have already been mapped, today mappers aren't as concerned with the accurate identification of the boundaries in their illustrations; they are looking more for the following characteristics within these boundaries, however crudely drawn:

- Spatial distribution of customers or potential customers;
- Differences based on geography; and
- Geographic areas of need.

Within the healthcare and health insurance arena, that translates to the following business applications:

- Patient or member location;
- Geographic variations in medical outcomes and practice patterns;
- Provider supply and demand by region or area;
- Geographic access studies;
- Marketing studies that analyze geographic variations in patient flows; and
- Improved targeting and services.

The change in emphasis from identifying boundaries to identifying characteristics within boundaries can result in a false sense of the territory involved. In these instances new methods may need to be employed to identify and isolate these boundaries. The examples below illustrate this evolution and the SAS solution for Roanoke Island.

Examples in this paper were run with SAS 9.4 and SAS Enterprise Guide 7.1 in a gridded environment. The techniques presented in this paper are not platform specific.

HISTORY OF MAP MAKING

Map making is believed to have developed independently in several parts of the world. Pacific Islanders made stick charts for navigation. Coastal maps made of ivory were carved by early Eskimos. Incas built relief maps of stone and clay. Chinese literature references maps created on silk as early as the 7th century BC. Early maps:

- Depicted small areas (e.g. trade route, battle plan);
- Were pictorial;
- Had no rules of orientation; and
- Did not depict true scale relationships.

The oldest known map is from Babylonia and dates back to 2300 BC. Very early maps were often fairly simple sketches on clay tablets. Early Egyptians and Chinese also made extensive use of simple maps in the very early periods. These early maps tended to be hand-drawn, limited in scope, and limited in distribution. Figure 1 shows an example of an early Viking map that is Viking-centric.



Figure 1. Early Viking Map

Map making continued to be refined and eventually culminated in the development of latitude and longitude lines by Ptolemy, imposing mathematical rules on the art of cartography. Using these advances, Ptolemy created maps of many parts of the world including the first known world map in 150 AD as part of his *Geographica* scientific treatise. Figure 2 shows this map.



Figure 2. Ptolemy's Map of the World

In the 15th century several developments contributed to the advancement of cartography. While these advancements meant that maps were more readily available, they were often proprietary. These advancements included:

- Invention of the printing press in 1440 – printed with carved wooden blocks followed by copper etchings.
- Contact with the Americas and the Far East.
- Increase in public learning.
- Need for navigational aids:
 - Coastlines, islands, rivers
 - Compass lines.

In addition, to maps, the oldest surviving globe also dates to the 15th century. Martin Behaim created the Erdapfel (earth apple) in 1492, currently on display at the Nuremburg Museum in Germany. This globe did not yet show the Americas. The first known world map to show the Americas was created by Martin Waldseemüller in 1507.



Figure 3. Waldseemüller's World Map

MAPPING ROANOKE ISLAND

Roanoke Island is located off the northeastern coast of North Carolina in an area known as the outer banks.



Figure 4. Roanoke Island Location

The first known map of Roanoke Island was created by John White in 1585. White sailed with Richard Grenville to the shore of present-day North Carolina in 1585, acting as artist and mapmaker to the expedition. White was also part of the 1587 failed expedition known as the "Lost Colony." White's maps were the first detailed maps of a major portion of the east coast of America.



Figure 5. Virginia 1585.

Below is cartographer Jan Jansson's map of the American East Coast from 1637. Jansson based his map upon Johannes De Laet's map of 1630. This map is considered a source map for New England and the Northeast, and was first to name Manhattan, New Amsterdam, the North River (Hudson) and South River (Delaware), along with the first appearance of Massachusetts.

Although not the emphasis of the map, note the inclusion of Roanoke Island.



Figure 6. Jan Jansson: Nova Anglia Novum Belgium et Virginia, 1637

In 1862, the following map was published providing information about the Battle of Roanoke Island during the Civil War. The battle was from February 7-8, 1862. Notice the original inlet was already closed.




Figure 7. Map Depicting Civil War Battle of Roanoke Island

The following map from the United States Geological Service (USGS) shows Roanoke Island in 2000, but again this is a map and service that is concerned with landform boundaries rather than the display of demographics or other analysis data. It does show the location of the old Roanoke inlet that has closed.




Figure 8. USGS Map of Roanoke Island

But as maps began to be generated by computer, the accuracy of coordinates was improved at the expense of the detail provided by the cartographer. Maps were output by line printers. Coordinates were stored as data. These early computer maps were dependent upon memory size and graphics capabilities. The advantage was in the display of multiple layers of information upon a single base. As base maps of North Carolina began to be developed by computer, Roanoke Island started disappearing. See Figures 7 and 8 below. Figure 7 is a map that is part of on-line learning about the state for elementary school students across North Carolina. Figure 8 is from the current state tourism information web site.




Figure 9. Map for Elementary School Students



Figure 10. North Carolina Tourism Map

SAS software was able to create boundary maps with SAS/GRAPH, a product introduced in 1980. Roanoke Island quickly disappeared from North Carolina in SAS maps also. Figure 9 shows a county map of North Carolina in a 1994 presentation by SAS, promoting SAS/GIS.




Figure 11. North Carolina Counties

This disappearance is not limited to the past. Figure 10 shows a map included in a SAS Global Forum 2016 presentation by SAS, discussing new features in SAS/ETS. Note that in Figure 10, the coastline just gets wrapped into the state and the Outer Banks completely disappear.




Figure 12. Weekly Car Sales in North Carolina by County

Research was conducted to see if these maps were the result of density reductions. But even more discouraging was the output from the SAS U.S. County map (maps.uscounty SAS map data set) with the maximum density. As illustrated in Figure 11 below, Roanoke Island just isn't there. Even worse, the maps.us map SAS map dataset eliminates the Outer Banks entirely as illustrated above.




Figure 13. Map using SAS USCOUNTY Dataset

THE SAS SOLUTIONS

While SAS may be producing maps that are part of the disappearance of Roanoke Island, fortunately SAS also provides solutions to include this island in SAS output. Two solutions are discussed here. 1) The first uses SAS to bring in a map as an external image and add to a SAS report. 2) The second uses the coordinates provided in the SAS map datasets to isolate and map Roanoke Island.

EXTERNAL IMAGE SOLUTION

SAS provides several means for adding images to reports and graphics. The most commonly used are the PREIMAGE and POSTIMAGE style attributes. These options are especially useful when wanting the image to be part of a table or graph. Other options available include using the MSOffice2K_x tagset, using the SAS Data Step Graphics Interface (DSGI), using the IBACK goption (background image), using the IFRAME option in Proc GCHART, using the IMAGE option on PATTERN statements and many more.

The option used in this solution is the Annotate IMAGE function. This solution is selected to provide more control over location. To display the image, coordinates are set with X and Y variables. The IMAGE function is selected. The current X and Y position provides one corner of the image while the opposite corner position follows the IMGPATH variable. The image file is specified in quotation marks on the IMGPATH variable.

In this example, Proc REPORT is used to create a table containing population information over time for the two towns on Roanoke Island. Below this report, the Annotate IMAGE function is used to load a Google map of the island to complete the report. The map shows the location of the towns. Census data for the population of the island as a whole is only available for the year 2000 and at that time more people lived on Roanoke Island outside the town boundaries than within. Due to lack of data, total population was not included. Because of the coastal location the Seaside style was selected for the report. Figure 12 shows this report.




Figure 14. Google Map of Roanoke Island

Solution One SAS Code:

Proc GPLOT is used to place the map image below the Proc REPORT table in this report example. Proc GPLOT was selected because this procedure gives total control to the programmer for placement of the image. To create this graphic, the normal plot axes, values and symbols are suppressed. The image is matched to a created data point in an annotate dataset using the image value of the annotate function variable. When using images, the most important change from using symbols or fonts is that it is necessary to create both horizontal and vertical space for the image. In this example, the space is created by expanding the axes and by adding to the x and y variables after reading in the image.

```
/*Create base dataset for plot*/
data imag_prep;
  t=1; l=1;run;

/*Create annotate dataset with image*/
data image;
  length function $ 8 style $ 15 imgpath $200.;
  retain xsys ysys '2' hsys '3' when 'a' position '4' style 'fit';
  set imag_prep;
  x=t-1; y=l-1;
  function='move'; output;
  imgpath='/cma/nshare/nophi/AA92483/A_2016Jan/roanoke_island.bmp';
  x=x+1; y=y+1;
  function='image'; output; run;

/*Set options for report*/
options nodate nonumber orientation=landscape leftmargin=".75in"
rightmargin=".75in" topmargin=.75in bottommargin=.25 in missing=' ';
goptions device=SASPRTC;
ods _ALL_ close;
ods escapechar='^';
ods pdf file="/cma/nshare/nophi/AA92483/sesug2016_1.pdf" notoc
style=seaside;
```



```

/*Create table with Proc Report*/
Proc REPORT data=roa_isl_pop missing nowd style(header)={fontsize=6}
style(column)={fontsize=5};
  column city ("Year" year,(population));
  define city / ' ' group width=20 style=[cellwidth= 2 in];
  define year / ' ' across style=[cellwidth= 1.2 in];
  define population/' ' analysis mean format=comma9.;
  title h=10 'Roanoke Island Cities';run;

/*Read in map and plot with GPLOT and annotate*/
ods pdf startpage=no;
goptions vsize=5; title ' '; symbol value=none;
axis1 label=none minor=none major=none style=0 order=(0 to 1 by .5)
value=none;
axis2 label=none minor=none major=none style=0 order=(0 to 1 by .5)
value=none;

Proc GPLOT data=imag_prep;
plot level*timeperiod/annotate=image
  haxis=axis1 vaxis=axis2 noframe;
run; quit;
ods pdf close;

```

SAS MAP DATASET SOLUTION

SAS examples often 1) use datasets with reduced boundary points, 2) reduce the density of maps in displayed output and 3) recommend the use of Proc GREduce to process map datasets so that they draw simpler maps with fewer boundary points. Emphasis has moved to the data displayed rather than the base maps. SAS maps still provide a means to a solution; the SAS map library datasets do include maps with the detail necessary to map small geographic entities like Roanoke Island.

SAS provides two types of SAS datasets for mapping. Traditional map datasets have been available with SAS/GRAPH since the product was released. These maps continue to be updated and enhanced. Since SAS 9.3, digital, vector-based base maps supplied by GfK Geomarketing are also available for use. While both types of maps continue to be updated, the GfK maps are intended to eventually replace the traditional maps. Advantages include a single source for map data throughout the world and increased resolution for many boundaries. Use the correct SAS map library to access these map datasets.

- MAPSGFK points to the GfK Geomarketing map datasets.
- MAPSSAS points to the traditional SAS boundary file.
- Unless updated to point to the GfK maps, MAPS refers to the MAPSSAS datasets.

The solutions below show maps with both of these source datasets, adding town/city names from the US cities dataset included with both map types. First is the preferred GfK map solution, followed the SAS traditional map dataset solution. The steps for both of these solutions are as follows:

- Isolate coordinates for county containing Roanoke Island – Dare County, NC.
- Locate centroid coordinates for Roanoke Island.
- Identify segment that includes Roanoke Island – Segment 4.
- Locate the minimum and maximum coordinates for this segment.
- Set up annotate file with the cities dataset to locate and label towns on the island.
- Use the minimum and maximum points to subset the cities annotate dataset.
- Select ODS output destination.
- Map the data.

Figure 13 shows the map of Dare County using the GfK map datasets. Figure 14 shows the subset of this map for Roanoke Island with the added city names. The datasets used here are the us_counties and uscity from the GfK library. These are projected datasets.




Figure 15. GfK Map of Dare County, NC.



Figure 16. GfK map of Roanoke Island, NC.

Solution Two GfK Map SAS Code:

```
/*Read in Dare County, NC from mapsgfk counties map dataset*/
data ncdare;
    set mapsgfk.us_counties(where=(fipstate(state)="NC" and county=055));
    x=long; y=lat; run;

/*County Map GMAP code*/
filename odsout '/cma/nshare/sesug/';
ODS LISTING CLOSE;
ODS HTML path=odsout body="daregfk1.htm" style=sasweb;
pattern1 v=s c=bibg;
Proc GMAP map=ncdare data=ncdare;
    id id;choro state/discrete nolegend; run; quit;
ods html close;
ods listing;

/*Isolate Roanoke Island*/
Proc MEANS data=ncdare min max;
    class segment;
    var x y;
    output out=boundary min=xmin ymin max=xmax ymax; run;
```

```

/*Find segment(s) for Roanoke Island using known central coordinates*/
data boundary; set boundary(where=(xmin lt -75.6615 and xmax gt -75.6615
and ymin lt 35.8897 and ymax gt 35.8897)); run;

/*Store macro values to isolate island using last row of values*/
data _null_;
set boundary;
call symput('xmin',xmin);call symput('xmax',xmax);
call symput('ymin',ymin);call symput('ymax',ymax);
call symput('seg',segment);run;

/*Create annotate dataset to label towns*/
data anno(where=(x ge &xmin and x le &xmax and y ge &ymin and y le
&ymax));
set mapsgfk.uscity(where=(statecode="NC" and county_name="Dare"));
length function $ 8; x=long; y=lat;
xsys='2'; ysys='2'; hsys='3'; when='a';
function='pie'; style='psolid'; rotate=360; size=1;
color='red'; output;
function='label'; position='9';
style=' '; rotate=.; size=2.5; color=' ';
text=trim(left(city)); output; run;

/*Map the Data*/
filename odsout '/cma/nshare/sesug/';
ODS LISTING CLOSE;
ODS HTML path=odsout body="roaislgfk2.htm" style=sasweb;
pattern1 v=s c=bibg;
Proc GMAP map=ncdare(where=(segment=4)) data=ncdare(where=(segment=&seg))
anno=anno;
id id;choro state/discrete nolegend; run; quit;
ods html close;
ods listing;

```

Figure 15 shows the map of Dare County using the SAS traditional map datasets. Figure 16 shows the subset of this map for Roanoke Island with the added city names. The datasets used here are the counties and uscity from the MAPSSAS data library. Projected datasets such as the states and uscounties datasets were tested but even at maximum density there is no Roanoke Island so an added step of projecting the datasets was needed for this output.




Figure 17. SAS traditional Map of Dare County, NC.




Figure 18. SAS traditional Map of Roanoke Island, NC.

Note that the resulting map from the traditional SAS map datasets has a much less refined boundary. While the GfK map has 104 segments for Dare County, the traditional map has 5. Also, because of fewer points in the dataset, the solution did not automatically remove Nags Head from the town/city labels. This of course can be done manually. The traditional SAS supplied cities data also provide fewer place names as illustrated by the two Roanoke Island maps.

Here is the code for the traditional SAS map dataset solution. Changes include the addition of map projection, the conversion of base coordinates to radians for locating center of Roanoke Island. While this is a usable solution, the GfK map would often be preferred for the greater detail in mapping small places.

Solution Two GfK Map SAS Code:

```
/*Subset Dare county and create keep variables for unprojected
coordinates*/
data dcounty;
    set maps.counties(where=(state=37 and county=55));
    ux=x; uy=y; run;

/*Map Dare County*/
Proc GPROJECT data=dcounty out=dare;
    id state; run;quit;
filename odsout '/cma/nshare/sesug/';
ODS LISTING CLOSE;
ODS HTML path=odsout body="dare_cntystd.htm" style=sasweb;
Proc GMAP map=dare data=dare;
    id county;
    choro state/nolegend;
run;quit;
ods html close;
ods listing;

/*Isolate Roanoke Island*/
Proc MEANS data=dare min max;
    class segment;
    var ux uy;
    output out=boundary min=xmin ymin max=xmax ymax; run;

/*Find segment(s) for Roanoke Island using known central coordinates in
radians*/
data boundary;
    set boundary(where=(xmin lt 1.320542292 and xmax gt 1.320542292
and ymin lt .6263934326 and ymax gt .6263934326)); run;

/*Store macro values to isolate island using last row of values*/
data _null_;
```



```

set boundary;
  call symput('xmin',xmin);call symput('xmax',xmax);
  call symput('ymin',ymin);call symput('ymax',ymax);
  call symput('seg',segment);run;

/*add cities*/
data annoc(where=(x ge &xmin and x le &xmax and y ge &ymin and y le
&ymax));
  set maps.uscity(drop=state where=(statecode="NC" and county=55));
  length function $ 8;
  x=long*arccos(-1)/180; y=lat*arccos(-1)/180; state=1000;
  xsys='2'; ysys='2'; hsys='3'; when='a';
  function='pie'; style='psolid'; rotate=360; size=1;
  color='red'; output;
  state+1;
  function='label'; position='9';style=' ';
  rotate=.; size=2.5; color='black';text=trim(left(city)); output; run;

/*Project data together*/
data all;
  set maps.counties(where=(state=37 and county=55)) annoc;run;

Proc GPROJECT data=all out=allp;
  id state;run;quit;
data citiessp darep;
  set allp;
  if state >100 then output citiessp;
  else output darep;run;

/*Map the data*/
filename odsout '/cma/nshare/sesug/';
ODS LISTING CLOSE;
ODS HTML path=odsout body="dare_std.htm" style=sasweb;
pattern1 v=solid color=cream;
Proc GMAP map=darep(where=(segment=4)) data=darep(where=(segment=&seg))
anno=citiessp;
  id county;
  choro state/nolegend;
run;quit;
ODS HTML CLOSE;
ODS LISTING;

```

CONCLUSION

This paper looks at the use of maps to set boundaries, display, analyze and communicate information. Over time maps have changed from emphasis on the accuracy of the geographic boundaries to emphasis on displaying a message, sometimes at a cost of the accurate boundaries. Roanoke Island was accurately mapped by Europeans explorers in the 16th century but has disappeared from many maps today.

Using Roanoke Island as the example, this paper illustrates how SAS offers two solutions to mapping geographic entities that don't represent commonly mapped areas such as states and counties. The first is incorporating maps from other sources into SAS output and the second is identifying and mapping the boundaries of these areas using SAS software. The second solution shows that with the inclusion of GfK map datasets, SAS provides a more accurate and consistent dataset from which to subset small areas than was available from the traditional SAS map datasets.

REFERENCES

- Aber JS. 2008. Brief History of Maps and Cartography. Emporia State University.
- Anzlic Committee on Survey and Mapping. Fundamentals of Mapping. <http://www.icsm.gov.au/mapping/history.html>. Accessed Sept 5, 2016.
- Bulkley J, Johnson S. "SAS/GIS Software: A Map is Worth a Thousand Pictures," SUGI 94 Proceedings. SAS Institute, Inc.
- Chovsta J. "Spatial Dependence, Nonlinear Panel Models, and More New Features in SAS/ETS® 14.1," <http://support.sas.com/resources/papers/proceedings16/SAS4101-2016.pdf>. Accessed Sept 7, 2016.
- Crowley J. A New and Correct Map of the Province of North Carolina drawn from the Original of Colo. Mosely's. 1737. North Carolina State Archives. <http://dc.lib.unc.edu/cdm/ref/collection/ncmaps/id/1245>. Accessed July 5, 2016.
- Jansson J. Nova Anglia Novum Belgium et Virginia, 1637. Barry Lawrence Ruderman Antique Maps, Inc. https://www.raremaps.com/gallery/detail/44935/Nova_Anglia_Novum_Belgium_et_Virginia/Jansson.html. Accessed Sept 8, 2016.
- Map of Roanoke Island, 1862. <http://www.learnnc.org/lp/multimedia/9917>. Accessed June 24, 2016.
- Merriam DF 1996. Kansas 19th century geologic maps. Kansas Academy of Science, Transactions 99, p. 95-114.
- Native Heritage Project. "John White Map Chowan Fort Discovery – Analysis," <https://nativeheritageproject.com/2012/05/07/john-white-map-chowan-fort-discovery-analysis/>. Accessed Sept 2, 2016.
- North Carolina Map. http://www.hotelfun4kids.com/hotelandresort/ushotels/north_carolina.htm. Accessed June 27, 2016.
- Nyerges, T.L. 1993. "Understanding the scope of GIS: Its relationship to environmental modeling." In Goodchild, M.F., Parks, B.O. and Steyaert, L.T. (eds.), Environmental modeling with GIS, p. 75-93. Oxford Univ. Press, 488 p.
- The Outer Banks of North Carolina. Professional Paper 1177-B. U.S. Department of Interior, U.S. Geological Survey. 2000.
- The Development of the Printed Atlas. Part 2: Ptolemaic Atlases. <http://www.mapforum.com/02/ptolemy1.htm>. Accessed June 23, 2016.
- SAS/GRAPH 9.4 Reference Fourth Edition. <http://support.sas.com/documentation/cdl/en/graphref/67881/PDF/default/graphref.pdf>. Accessed June 23, 2016.

ACKNOWLEDGMENTS

Thanks go to the Roanoke Island Historical Association, the National Park Service and the Lost Colony Outdoor Drama for making available some of the information and sources used in this paper.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Barbara B. Okerson, Ph.D., CPHQ, FAHM
Health Information Manager
Health Care Analytics
Anthem, Inc.
barbara.okerson@anthem.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.